PWL_Noise_Ref_2_Input

```
* www.idea2ic.com dsauersanjose@aol.com  4/22/08
```



```
.OPTIONS  GMIN=1e-18    METHOD=trap   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
VCC       VCC    0             DC     1
VEE       0      VEE           DC     2
IB1       VB     0             DC    -1u
XQN1      VTA    0      VE     VC1    XQ1X
QN2       OUT    0      VE     NPN1   1.00
XQN3      VE     VB     VEE    VC2    XQ1X
QN4       VB     VB     VEE    NPN1   1
C1        VB     VEE    1p
QP1       VTA    VTA    VCC    PNP1   1
QP2       OUT    VTA    VCC    PNP1   1
R1        OUT    0                    1
VC1       VC1    0             DC     1
VC2       VC2    0             DC     0
Ioff      OUT    0             DC     0

Vstart    VS     0             Pulse  (0 1 1m 1n 1n 10s 20s)
B2        OUT2   0             V = v(OUT)*v(VS)+1p
XM_RMS    OUT2   V_AVE  V_RMS  M_RMS
.tran     5u                   50m        0          5u         UIC

**#0===See_The_Referred_To_Input_Noise_In_Action=====
.control
**#1===Verify_that_XQN1_has_The_Right_Noise_At_IB=1uA=====
run
plot     out2 v_ave v_rms                        title  IB_1uA
plot     ve    ylimit -577.61m -577.624m      title  IB_1uA

**#2===Verify_that_XQN1_has_The_Right_Noise_At_IB=100uA=====
alter      ib1               dc  =  -100u
run
plot     out2 v_ave v_rms                        title  IB_100uA

**#3===Measure Output Noise current XQN3 for @ IB = 100uA=====
alter      VC1               dc  =  0
alter      VC2               dc  =  1
run
plot     out2 v_ave v_rms                        title  IB_100uA_tail_Only

**#4====Offset QN2 by 1mV and re-measure Output Noise current==
alter      QN2               area  =  1.04
alter      Ioff              dc  =  -1.9608u
run
plot     out2 v_ave v_rms    ylimit -.03n .03n    title  IB_100uA_tail_Offset
.endc

*====== .models Using Near Perfect Transistors ============
.model    NPN1    NPN( BF = 1MEG VAF =   1MEG )
.model    PNP1    PNP( BF = 1MEG VAF =   1MEG )

*====== .SUBCKT  XQ1X Models Audio Noise to current ============
```



```
.SUBCKT  XQ1X  C   B   E    NCN
```

```
.include                          Noise_audio1K.txt
QN1          C      B1     E1     NPN1    1
VM           E1     E      DC     0
BN           B      B1     V =    v(NCN)*v(audio1K)*(13u/(i(VM)+5n))^.5
.ENDS        M_RMS

* ====== .SUBCKT M_RMS Sanity Check RMS Meter =======
*              ______
*             |      |-> V_AVE
*       V_IN->| RMS  |                 A Sanity RMS Meter
*             |______|-> V_RMS
*
*
*        Vsum     Rave          Vsum2    Rave2        V_RMS     Rrms
*        __/\  /\  /\_ V_AVE   __/\  /\  /\_ Vave2 __/\  /\  /\_
*      _|_    \/  \/   |     _|_    \/  \/   |     _|_    \/  \/  |
* V_IN->/   \          _|_   /   \          _|_   /   \           _|_
*       \___/Bsum      ___   \___/Bsum2     ___   \___/Brms       ///
*         |             |      |             |      |
*        _|_      Cave _|_    _|_     Cave2 _|_    _|_
*        ///           ///    ///           ///    ///
*
.SUBCKT M_RMS   V_IN   V_AVE V_RMS
Bsum        Vsum      0         V = v(V_IN)
Rave        Vsum      V_AVE     1
Cave        V_AVE     0         10m
Bsum2       Vsum2     0         V = (v(V_IN)-v(V_AVE))^2
Rave2       Vsum2     Vave2     1
Cave2       Vave2     0         7m
Brms        V_RMS     0         V = v(Vave2)^0.5
Rrms        V_RMS     V0        1k
.ENDS        M_RMS

.end

=====================END_OF_SPICE============================

**#1===First measure Output Noise current XQN1 @ IB = 1uA=====
run
plot      out v_ave v_rms                   title  IB_1uA
plot      ve    ylimit -577.61m -577.624m   title  IB_1uA

*       THE NOISE CURRENT MODEL
*
*                      ^  VCC
*                     / \
*           _________ _|_ __________
*          |                        |
*          |                        |
*           ->                     <-
*          QP1`|___________   |'QP2    -> Iout = In
*              '|  |      |   |`_
*        |   |_____|      |  |   |     |______|OUT|
*        V   |  VTA       V In   |            |___|
*            |          __|__    |        |
* (1uA-In)/2 |         /  _  \   |        V
*            |         \/ \/     | (1uA-In)/2
*            |   <===  /\_/\     |
*       ___|'XQN1      \___/    QN2 `|___
*      _|_  |`->   VE     |      <-'|   _|_
*      /// :  |__________|__________|    ///
*         VC1    _|_
*               /   \
*               \/ \/1uA    sqrt(2*1.6e-19*500n*20KHz)
*               /\_/\
*               \___/       Ishot = 56.6pA_rms (audio)
*                _|_
*                \ /
*                 V
*
```

Noise current across the emitter to collector follows the equation.



Noise can also be modeled as equivalent input noise voltage at the base

```
*         THE INPUT NOISE VOLTAGE MODEL
*
*                               ^   VCC
*                              /_\
*                        _______|_________
*                       |                 |
*                       |                 |
*   Ishot = 56.6pA_rms   ->             <-
*   R_gm  = 26mV/500nA  QP1`|_______  |'QP2   -> Iout = Ven/R_gm
*         = 52K         _  '|  |      |`_
*   Vshot = Ishot*R_gm  |______|        |_______|LP |
*         = 2.94uV_rms  | VTA           |       |___|
*                |      |                 ^
*              \ | /    |  |Ven/(2*R_gm)| |Ven/(2*R_gm)
*                V      |  V            | |
*                       |               |
*           __/Ven\___|'XQN1        QN2 `|__
*          _|_ \___/    |`->   VE        <-'| _|_
*          ///         :|_______________|    ///
*                    VC1   _|_         ^
*                         /   \       /|\
*                         \/ \/1uA     |
*                         /\_/\
*                         \___/      1.472uV_rms
*                          _|_       (voltage divide)
*                          \ /
*                           V
*
```



```
**#2===Measure Output Noise current XQN1 for @ IB = 100uA=====
alter      ib1              dc = -100u
run
plot      out v_ave v_rms                        title  IB_100uA
```

A factor of 100 should increase noise current by 10





All Matched pairs add noise.

Transistor QN2 will increase the input noise by a factor of sqrt(2).

Considering just two input transistors, the signal to noise is defined by Idc.

SHOT SIGNAL TO NOISE RATIO LIMIT

+/- Iout = 2*Idc(pk) = sqrt(2)*Idc(rms)

ShotNoise = sqrt(2)*sqrt(2*q*Idc*BW)(rms)

S/N = sqrt(Idc/(2*q*BW))



S/N = sqrt(50u/(2*1.6e-19*20kHz) = 99dB

```
**#3===Measure Output Noise current XQN3 for @ IB = 100uA=====
alter      VC1              dc = 0
alter      VC2              dc = 1
run
plot      out v_ave v_rms                    title  IB_100uA_tail_Only
```

A perfectly matched input stage will completely common mode that noise out.





```
**#4====Offset QN2 by 1mV and remeasure Output Noise current==
alter      QN2            area = 1.04
alter      Ioff           dc = -1.9608u
run
plot      out v_ave v_rms   ylimit -.3n .3n    title  IB_100uA_tail_Offset
```

A 1mV offset voltage is typical for a diff stage.
This is modeled by increasing QN2 to 1.04x.



Still the noise level is reduced such that the noise is still dominated by the two input transistors.



Only the Matched pair transistors add noise.